

Math 2280-2
 Tuesday February 20
 Solving DE's in Maple
 Resonance and beading

These notes discuss material from sections 3.5 and 3.6 of the text, in the context of Maple. They may be helpful when you try to do the Maple project, and are online at our Maple page, go to www.math.utah.edu/~korevaar/2280spring01, click on the Maple projects and handout link, and then on the 2280feb20.pdf or 2280feb20.mws links.

Part 0: checking homework problems on Maple

Here's a homework problem worked out in Maple, #37 from section 3.5. This problem was a pain to work by hand, in fact we did it on Friday. For more information on the commands which are used, use the Help files. I found the syntax for putting higher order derivatives and initial conditions into "dsolve" under the topic "dsolve IC".

```
> with(DEtools):
#DE library of commands
>
f:=x->1+x*exp(x);
#right hand side
deqtn:= diff(y(x),x,x,x) -2*diff(y(x),x,x)
+ diff(y(x),x) =f(x);
#differential equation from #37
```

$$f := x \rightarrow 1 + x e^x$$

$$deqtn := \left(\frac{\partial^3}{\partial x^3} y(x) \right) - 2 \left(\frac{\partial^2}{\partial x^2} y(x) \right) + \left(\frac{\partial}{\partial x} y(x) \right) = 1 + x e^x$$

```
> ans37:=dsolve({deqtn,y(0)=0,D(y)(0)=0,D(D(y))(0)=1},y(x));
#getting the expression for second derivative right
#took a little fiddling. But it still only took a few
#minutes.
```

$$ans37 := y(x) = \frac{1}{6} x^3 e^x - \frac{1}{2} x^2 e^x + 3 x e^x - 4 e^x + x + 4$$

Part 1: Undamped, forced oscillators, w and w0 different: If the forcing frequency is not equal to the natural frequency the general solution to will be the superposition of two cos-sin terms, one corresponding to the particular solution with angular frequency w, and the other being the general solution to the homogeneous problem, with angular frequency w0. When w and w0 as well as the corresponding amplitudes are close, the system exhibits **beating**. Musicians tune their instruments using this phenomenon.

```
> restart:with(DEtools):with(plots):
Warning, the name changecoords has been redefined
> deqtn1a:=diff(x(t),t,t) + w0^2*x(t) = (F0/m)*cos(w*t);
#this is the case w not equal to w0,
#I divided the model equation by m
```

$$\text{deqtn1a} := \left(\frac{\partial^2}{\partial t^2} x(t) \right) + w_0^2 x(t) = \frac{F_0 \cos(w t)}{m}$$

```
> dsolve(deqtn1a,x(t));
#general solution, equal to particular
#solution plus general homogeneous eqtn solution.
```

$$x(t) = _C2 \cos(w_0 t) + _C1 \sin(w_0 t) - \frac{F_0 \cos(w t)}{m(-w_0^2 + w^2)}$$


```
> sol4:=dsolve({deqtn1a,x(0)=0,D(x)(0)=0},x(t));
#a nice choice of initial conditions
```

$$\text{sol4} := x(t) = \frac{F_0 \cos(w_0 t)}{m(-w_0^2 + w^2)} - \frac{F_0 \cos(w t)}{m(-w_0^2 + w^2)}$$

```
> sol5:=subs({w=45,w0=55,F0=50,m=.1},sol4);
#Example 2 page 210
```

$$\text{sol5} := x(t) = -.5000000000 \cos(55 t) + .5000000000 \cos(45 t)$$

```
> actual:=plot(rhs(sol5),t=0..3,color=black):
envel1:=plot(sin(5*t),t=0..3,color=black):
envel2:=plot(-sin(5*t),t=0..3,color=black):
display({actual,envel1,envel2},title=
"Figure 3.6.3 page 210");
```



Explanation of Beating: Beating is a superposition phenomenon. You observe it in many vibration problems. Let

```
> restart:
```

```
f:=t->C0*cos(w0*t - a);
g:=t->C1*cos(w1*t-b);
```

$$f := t \rightarrow C0 \cos(w0 t - a)$$

$$g := t \rightarrow C1 \cos(w1 t - b)$$

Then beating has to do with the difference (or sum) of f and g. To be consistent with the example above, we consider the difference

```
> h:=g-f;
h(t);
```

$$h := g - f$$

$$C1 \cos(w1 t - b) - C0 \cos(w0 t - a)$$

Then if the two amplitudes C0 and C1 are close, and if the two frequencies w0 and w1 are close, there will be long time intervals in which h=g-f essentially equals twice one of them, followed by long time intervals in which their difference is essentially zero. In the example above, C0=C1=1/2 and a=b=0.

In fact, if we set a=b=0 and C1=C2=C, then we get

```
> h1(t):=t->C*(cos(w1*t)-cos(w0*t));
```

$$h1(t) := t \rightarrow C (\cos(w1 t) - \cos(w0 t))$$

Mathematically, beating is explained by defining the average angular frequency $w=(w0 + w1)/2$, and half the difference $dw=(w1-w0)/2$, so that h1(t) is given by

```
> h1:=t->C*(cos((w+dw)*t)-cos((w-dw)*t));
```

$$h1 := t \rightarrow C (\cos((w + dw) t) - \cos((w - dw) t))$$

which we can expand using the cosine addition angle formula:

```
> expand(h1(t), trig);
```

$$-2 C \sin(t w) \sin(t dw)$$

So we see that this difference of cosines is actually a product of sines. If dw is small, then we can consider $2C\sin(t*dw)$ as a slowly varying amplitude (with large period $T=2*\text{Pi}/dw$), and this will create the "envelope" for the more rapidly oscillating $\sin(tw)$ term. In our example the envelope was created by $\sin(5t)$, and the rapid oscillations were created by $\sin(50t)$.

Part 2: Resonance in undamped, forced harmonic oscillators, when $w=w0$: In class we used the method of undetermined coefficients to solve the forced oscillator, with no damping, in the case that the driving frequency w exactly equals the natural frequency w0. Well, we found a particular solution. We were considering the DE

```
> deqtn1:=diff(x(t),t,t) + w0^2*x(t) = (F0/m)*cos(w0*t);
#I have again written k/m = w0^2
```

$$deqtn1 := \left(\frac{\partial^2}{\partial t^2} x(t) \right) + w0^2 x(t) = \frac{F0 \cos(w0 t)}{m}$$

```
> sol1:=dsolve(deqtn1,x(t));
#general solution
```

$$sol1 := x(t) = _C1 \sin(w0 t) + _C2 \cos(w0 t) + \frac{\frac{1}{2} F0 t \sin(w0 t)}{w0 m} + \frac{\frac{1}{2} F0 \cos(w0 t)}{m w0^2}$$

You see there the particular solution which we found, namely xp(t):

```
> xp:=t->(1/2)*F0*t*sin(w0*t)/(w0*m);
```

$$xp := t \rightarrow \frac{1}{2} \frac{F0 t \sin(w0 t)}{w0 m}$$

If we wanted to solve the general initial value problem, we would have set things up this way:

```
> sol2:=dsolve({deqtn1, x(0)=x0, D(x)(0)=v0}, x(t));
simplify(rhs(%));
```

$$sol2 := x(t) = \frac{v0 \sin(w0 t)}{w0} - \frac{1}{2} \frac{(F0 - 2 x0 m w0^2) \cos(w0 t)}{m w0^2} + \frac{\frac{1}{2} F0 t \sin(w0 t)}{w0 m} + \frac{\frac{1}{2} F0 \cos(w0 t)}{m w0^2}$$

$$\frac{\frac{1}{2} \frac{2 v0 \sin(w0 t) m + 2 \cos(w0 t) x0 m w0 + F0 t \sin(w0 t)}{w0 m}}$$

You can see from this form of the solution that xp(t) solves the initial value problem x0=v0=0, i.e. the system initially at rest. The other two terms give the effects of non-zero x0 and v0, respectively.

We can now plug in any values for the various parameters, using the "subs" command. For example, suppose F0=1, m=1, w0=3, x0=0 and v0=0. We could dsolve with those particular values, or we can plug them into sol2:


```
> sol3:=subs({F0=1, m=1, w0=3, x0=0, v0=0}, sol2);
```

$$sol3 := x(t) = \frac{1}{6} t \sin(3 t)$$

```
>
```

Graphing solutions from dsolve: If we want to graph this x(t) we cannot just try to plot x(t), because it hasn't actually been defined yet. We have to extract it from the right hand side of sol3. One way is manually type in the formula. Or, use your mouse to highlight it (in this case the 1/6*t*sin(3*t)) and then paste it into the "plot" command. Or, plot "rhs(sol1)".

```
> with(plots):
#plotting library
> plot(rhs(sol3), t=0..15, color=black);
#A classic picture of resonance.
```



The Transition from beating to resonance. If the driving frequency ω_1 is allowed to vary, and approaches the natural frequency ω_0 , then the beating packets will stretch out since $T=2\pi/d\omega$ will approach infinity. The beating picture will transform into the resonance picture. For example, consider the following initial value problem, which is like the one we just , except that the driving angular velocity is 3.1 instead of 3:

```

> restart:with(DEtools):with(plots):
Warning, the name changecoords has been redefined
> deqtn:=diff(x(t),t,t) + 9*x(t)=cos(3.1*t):
dsolve({deqtn,x(0)=0,D(x)(0)=0});


```

$$x(t) = -\frac{100}{61} \cos\left(\frac{31}{10}t\right) + \frac{100}{61} \cos(3t)$$

```


> plot(-100/61*cos(31/10*t)+100/61*cos(3*t),
t=0..15, color=black);
#this will look very close to previous picture

```



This looks like resonance, like the picture we just made. But on a longer time scale we see that it is actually beating, with the beating period equal to:

```
[ > evalf(2.0*Pi/(0.05));
                                125.6637062
> plot(-100/61*cos(31/10*t)+100/61*cos(3*t),
      t=0..62.8, color=black);
#half(?) a beating period
```



Explanation: Mathematically, if we consider the undamped spring with natural frequency w_0 , then from our work in example 2 we know the solution to the initially at rest initial value problem is

```
[ > restart:
  x1:=t->(F0/(2*m*w0))*t*sin(w0*t);
                                
$$x1 := t \rightarrow \frac{1}{2} \frac{F0 t \sin(w0 t)}{m w0}$$

```

[And for $w=w_0+dw$ the corresponding solution (from example 3) is

```
[ > x2:=t->(F0/(m*((w0+dw)^2 - w0^2)))*(cos(w0*t) - cos((w0+dw)*t));
      x2:=t->
$$\frac{F0(\cos(w0 t) - \cos((w0 + dw) t))}{m((w0 + dw)^2 - w0^2)}$$

```

If you expand x2(t), first using cos addition, and then doing a MacLaurin approximation (Taylor), about dw=0, you get an approximation for x2(t) which will be valid as long as (dw)*t is small. This is a good exercise to do by hand. Here's the answer from Maple:

```
[ > simplify(x2(t), trig);
      
$$\frac{F0(\cos(w0 t) - \cos((w0 + dw) t))}{m dw (2 w0 + dw)}$$

[ > expand(%, trig);
      
$$\frac{F0 \cos(w0 t)}{m dw (2 w0 + dw)} - \frac{F0 \cos(w0 t) \cos(t dw)}{m dw (2 w0 + dw)} + \frac{F0 \sin(w0 t) \sin(t dw)}{m dw (2 w0 + dw)}$$

[ > taylor(%, dw=0, 2);
      #second arg says expand about dw=0,
      #third arg says compute expansion through
      #first order
      
$$\frac{1}{2} \frac{F0 \sin(w0 t) t}{m w0} + O(dw)$$

```

(When I try taylor right away on x2(t) Maple gives me an error message, because a dw in the denominator has to be canceled with a dw in the numerator, before setting dw=0; I don't know why first simplifying and then expanding let's Maple see what to do, but there are a lot of things in life that I don't understand.) The final answer shows that for small dw (and small dw*t), x1(t) is the first order approximation for x2(t). In other words the beating picture looks like the resonance picture, for t in a fixed interval.

```
[ >
```

Part 4: Forcing damped harmonic oscillators. When $c > 0$ (damping), the homogeneous part of the general solution dies out, no matter what the initial conditions are. Therefore this part of the solution is called TRANSIENT. The part of the solution which is a linear combination of $\cos(\omega t)$ and $\sin(\omega t)$ (and which we get by the method of undetermined coefficients as a particular solution), persists, and is called the STEADY STATE solution. No matter what the initial conditions are, you will always end up with the same steady state solution for a given spring system! (WHY?) Let's study these phenomena, by doing Example 6, page 215.

```
[ > restart:with(DEtools):with(plots):
Warning, the name changecoords has been redefined
[ > deqtn:=diff(x(t),t,t) +2*diff(x(t),t) + 26*x(t) = 82*cos(4*t);
      dsolve({deqtn, x(0)=6,D(x)(0)=0});
      
$$deqtn := \left( \frac{\partial^2}{\partial t^2} x(t) \right) + 2 \left( \frac{\partial}{\partial t} x(t) \right) + 26 x(t) = 82 \cos(4 t)$$

```

$$x(t) = 5 \cos(4t) - 3 e^{(-t)} \sin(5t) + 4 \sin(4t) + e^{(-t)} \cos(5t)$$

So we see the transient solution (the terms with $\exp(t)$ in them), and the steady periodic solution. We could convert the steady state solution into $C\cos(4t-a)$, if we wanted to, as is done on page 348. For our graph it is enough to notice that the amplitude of the steady state solution is


```
> C:=sqrt(5^2+4^2);
evalf(%);
```

$$C := \sqrt{41}$$

$$6.403124237$$

Here's a plot of the solution, decomposed into transient and steady state pieces. Notice we got the steady-state amplitude right.

```
> xtrans:=t->-3*exp(-t)*sin(5*t)+exp(-t)*cos(5*t):
xsp:=t->5*cos(4*t)+4*sin(4*t):
plot1:=plot(xtrans(t),t=0..4,color=black):
plot2:=plot(xsp(t),t=0..4,color=black):
plot3:=plot(xtrans(t)+xsp(t),t=0..4,color=black):
display({plot1,plot2,plot3});
```



Practical resonance. A damped system can't resonate completely, but different driving frequencies will effect the amplitude of the steady state solution. For small "c" one expects the steady-state amplitude to get large with driving frequency near to that of the natural frequency of the system. This principal can be a useful tool in engineering: find the natural frequencies of systems in the theoretical case of $c=0$, and be sure to damp frequencies near those in the actual physical problem. For our forced oscillator it is not too hard to calculate the amplitude of the steady state solution, as the text does on page 346. Doing this by hand is a good exercise; you may get more out of it than you get out of letting Maple do the work:

```
> restart:with(linalg):with(plots):
```


Warning, the protected names norm and trace have been redefined and unprotected

Warning, the name changecoords has been redefined

```
> xp:=t->A*cos(w*t) + B*sin(w*t);
```

$$xp := t \rightarrow A \cos(w t) + B \sin(w t)$$

```
> eqtn1:=m*difff(xp(t),t,t)+c*difff(xp(t),t) + k*xp(t)=F0*cos(w*t);
```

```
eqtn1 :=
```

$$m(-A \cos(w t) w^2 - B \sin(w t) w^2) + c(-A \sin(w t) w + B \cos(w t) w) + k(A \cos(w t) + B \sin(w t)) = F0 \cos(w t)$$

Equating coefficients for coswt and sinwt leads to the system:

```
> what:=solve({-m*A*w^2 + c*B*w + k*A = F0,
               -m*B*w^2 -c*A*w + k*B = 0}, {A,B});
```

$$what := \left\{ B = \frac{c w F0}{c^2 w^2 + m^2 w^4 - 2 m w^2 k + k^2}, A = -\frac{(m w^2 - k) F0}{c^2 w^2 + m^2 w^4 - 2 m w^2 k + k^2} \right\}$$

```
> what[1];
```

$$B = \frac{c w F0}{c^2 w^2 + m^2 w^4 - 2 m w^2 k + k^2}$$

```
> rhs(what[1]);
```

$$\frac{c w F0}{c^2 w^2 + m^2 w^4 - 2 m w^2 k + k^2}$$

```
> amplitude:=simplify(sqrt(rhs(what[1])^2 + rhs(what[2])^2));
#this is the formula (21) on page 346.
```

$$amplitude := \sqrt{\frac{F0^2}{c^2 w^2 + m^2 w^4 - 2 m w^2 k + k^2}}$$

Continuing example 6 on page 347:

```
> examp6:=subs({F0=82,c=2,k=26,m=1},amplitude);
```

$$examp6 := \sqrt{6724} \sqrt{\frac{1}{-48 w^2 + w^4 + 676}}$$

```
> amp:=t->subs(w=t,examp6);
```

$$amp := t \rightarrow \text{subs}(w = t, \text{examp6})$$


```
> amp(t);
```

$$\sqrt{6724} \sqrt{\frac{1}{-48 t^2 + t^4 + 676}}$$

```
> plot(amp(w),w=0..20, color=black,
      title="figure 3.6.9 page 216");
#practical resonance, page 216
```

```
>
```

figure 3.6.9 page 216



```
> solve(diff(amp(w),w)=0,w);  
    #find w value at max amplitude  
                                0, 2*sqrt(6), -2*sqrt(6)  
> evalf(2*sqrt(6));evalf(amp(2*sqrt(6)));  
    #practical resonance  
                                4.898979486  
                                8.200000000  
> evalf(sqrt(26));  
    #natural frequency of undamped system  
                                5.099019514
```

END!